Includes barcode for easy software entry.

HEWLETT-PACKARD

HP-41

USERS' LIBRARY SOLUTIONS

Business Stat/Marketing/Sales



NOTICE

The program material contained herein is supplied without representation or warranty of any kind. Hewlett-Packard Company therefore assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press [■] [GTO] [.] [.] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⋇. When you see ⋇ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press [■] [APPEND] in ALPHA mode (press [■] and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: [■], followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [.] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: [■] [.] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing [■] and then the indirect address. Stack addresses are specified by pressing [.] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing [■] [.] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01◆LBL "SAM	[■] [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL"SAMPLE*
PLE"	[ALPHA] *THIS IS A* [ALPHA]	*02"THIS IS A*
02 "THIS IS	[ALPHA] [■] [APPEND] SAMPLE	*03" ⊢ SAMPLE*
A "	[■] AVIEW [ALPHA]	*04 AVIEW*
03 "⊢SAMPLE	6	*05 6*
"	[ENTER↑]	*06 ENTER ↑*
04 AVIEW	2 [CHS]	*07 -2*
05 6	[÷]	*08 /*
06 ENTER↑	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
07 -2	[STO] [■] [.] L	*10 STO IND L*
08 /	[ALPHA] R3= [■] [ARCL] 03	*11"R3=*
09 ABS	[■] [AVIEW]	*12 ARCL 03*
10 STO IND	[ALPHA]	*13 AVIEW*
L	[■] [RTN]	*14 RTN*
11 "R3="		
12 ARCL 03		
13 AVIEW		
14 RTN		

TABLE OF CONTENTS

FORECASTING USING EXPONENTIAL SMOOTHING

Exponential smoothing is a special kind of moving average. It is often used for short-term sales and inventory forecasts. Typical forecast periods are monthly or quarterly.

Unlike a moving average, exponential smoothing does not require a great deal of historical data. This program, for example, forecasts demand by using only a smoothing constant, an "old smoothed average," and a current-period usage statistic.

This program is a singly-smoothed exponential forecasting routine which: (1) accomodates quarterly seasonal correction factors, (2) can handle some trend in the data, (3) produces smoothed estimates of current demand, D_t, (4) produces next-period smoothed demand estimates, D_{t+1}, (5) calculates a mean absolute deviation, MD, and a tracking ratio, T, (6) provides a goodness of fit measure, V, which measures the variance between the next period's demand estimate to that period's actual demand, and (7) provides for convenient restarting when the user wishes to update a data series.

This program should not be used with data which has more than a moderate amount of up or down trend. And, at least two projections of D_{t+1} must be done before MD or T can be calculated.

Counter Q_i is used with the deseasonalizing adjustment option. Q_i keeps track of the fiscal quarter or calendar quarter associated with a given X_t, and should always be for time period t-1.

Equations:

α = smoothing constant ($0 < \alpha < 1$)

X_t = actual current period usage smoothed average,

$S_t = \alpha X_t + (1-\alpha)S_{t-1}$

change, $C_t = S_t - S_{t-1}$

trend, $T_t = \alpha C_t + (1 - \alpha)\, T_{t-1}$

current period expected usage, $D_t = S_t + \frac{(1 - \alpha)}{\alpha} T_t$

forecase of next period expected usage, $D_{t+1} = S_t + (\frac{1}{\alpha})\, T_t$

error, $e = D_t - X_t$

cumulative error = $\sum_{t-1}^{m} e^2$

initial conditions: $S_{t-1} = X_{t-1}$

$T_{t-1} = 0$

SV = seasonal variation factor

References:

HP-67/97 Users' Library program #01206D written by Professor Robert Olsen.

Robert Goodell Brown, Smoothing, Forecasting, and Prediction of Discrete Time Series, Englewood Cliffs, New Jersey: Prentice-Hall, 1963.

Elwood S. Buffa and William H. Taubert, Production-Inventory Systems: Planning and Control, Rev. ed., Homewood, Illinois: Richard D. Irwin, 1972.

Norbert Lloyd Enrick, Market and Sales Forecasting, San Francisco, California: Chandler Publishing Co., 1969.

Example:

Test Data when deseasonalization done.

Time Period	Actual X_t
0	100
1	100
2	150
3	70

α = 0.2 Q_o=4, SV_1=1.15, SV_2=0.94, SV_3=0.89, SV_4=1.02

Keystrokes:	Display:
[USER]	(Set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 023	
[XEQ] [ALPHA] SMOOTH [ALPHA]	ALPHA=?
.2 [R/S]	0.00
1.15 [ENTER↑] .94 [ENTER↑]	
.89 [ENTER↑] 1.02 [///] [B]	1.02
4 [ENTER↑] 100 [A]	102.00
100 [B]	B=1
[R/S]	D1=106.68 (Deseasonalized smoothed D_t)
[R/S]	SD1=92.77 (Seasonalized smoothed D_t)
[R/S]	D2↑=107.20 (Deseasonalized forecast)
[R/S]	SD2↑=114.04 (Seasonalized forecast)

Keystrokes:	Display:
150 [B]	B=2
[R/S]	D2=119.37
[R/S]	SD2=126.99
[R/S]	D3↑=121.24
[R/S]	SD3↑=136.22
[C]	MD=33.80
[R/S]	T=1.00
[R/S]	Σe2=1142.44
[R/S]	B=2.00
[R/S]	V=571.22
70 [B]	B=3
[R/S]	D3=100.02
[R/S]	SD3=112.38
[R/S]	D4↑=99.54
[R/S]	SD4↑=97.58
[C]	MD=46.37
[R/S]	T=-0.54
[R/S]	Σe2=4616.36
[R/S]	B=3.00
[R/S]	V=1538.79

User Instructions

SIZE: 023

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program and set USER mode		[USER]	
2.	Initialize		[XEQ] SMOOTH	ALPHA=?
3.	Key in value of alpha	α	[R/S]	0.00
4.	(Optional) Key in seasonal coefficients	SV_1	[ENTER↑]	SV_1
		SV_2	[ENTER↑]	SV_2
		SV_3	[ENTER↑]	SV_3
		SV_4	[///] [B]	SV_4
5.	(Optional) Store T_{t-1} if known.	T_t	[STO] 04	T_{t-1}
6.	Key in Q_i for T_{t-1}, X_{t-1} (or S_{t-1} if known)	Q_i	[ENTER↑]	Q_i
		X_{t-1}	[A]	X_{t-1} (D)
7.	Key in X_t and calculate expected current			
	usage	X_t	[B]	$B=t_n$
			[R/S]	Dt_n =
8.	Calculate expected seasonalized current			
	usage		[R/S]	SDt_n =
9.	Calculate D_{t+1} and set up calculations for			
	MD, T and V		[R/S]	Dt_{n+1} =
			[R/S]	SDt_{n+1} =
10.	Calculate MD, T, Σe_i, V		[C]	MD=
			[R/S]	T=
			[R/S]	Σe2=
			[R/S]	B=
			[R/S]	V=
11.	Continue keying in data using steps 7-9,			
	repeating step 10 as often as desired.			

Program Listings

```
 01◆LBL "SMO
OTH"
 02 SF 21
 03 CF 29
 04 CLRG
 05 1
 06 STO 10
 07 STO 11
 08 STO 12
 09 STO 13
 10 "ALPHA=?
"
 11 PROMPT
 12 STO 18
 13 -
 14 STO 19
 15 CLX
 16 RTN
 17◆LBL b
 18 STO 13
 19 RDN
 20 STO 12
 21 RDN
 22 STO 11
 23 RDN
 24 STO 10
 25 RDN
 26 RTN
 27◆LBL A
 28 STO 20
 29 X<>Y
 30 STO 08
 31 9
 32 +
 33 RDN
 34 RCL IND
T
 35 *
 36 STO 21
 37 STO 01
 38 RTN
 39◆LBL B
 40 STO 00
 41 RCL 08
 42 4
 43 /
 44 FRC
 45 4
 46 *
 47 1
 48 +
 49 STO 08
```

Initialize

$\alpha \rightarrow R_{18}$

$1-\alpha \rightarrow R_{19}$

Enter SV's

Enter start-up data

Enter X_t

```
 50 9
 51 +
 52 RDN
 53 RCL IND
T
 54 *
 55 STO 22
 56 1
 57 ST+ 09
 58 "B="
 59 XEQ 02
 60 AVIEW
 61 1
 62 RCL 09
 63 X=Y?
 64 GTO 01
 65 RCL 22
 66 RCL 17
 67 -
 68 ST+ 14
 69 STO 16
 70 ABS
 71 ST+ 15
 72 X↑2
 73 ST+ 03
 74◆LBL 01
 75 RCL 19
 76 RCL 01
 77 *
 78 RCL 18
 79 RCL 22
 80 STO 21
 81 *
 82 +
 83 STO 02
 84 RCL 01
 85 -
 86 RCL 18
 87 *
 88 RCL 04
 89 RCL 19
 90 *
 91 +
 92 STO 04
 93 RCL 19
 94 RCL 18
 95 /
 96 *
 97 RCL 02
 98 STO 01
 99 +
100 STO 06
```

Prior period estimate $D_{t+1(I)}$

Σe_i^2

Calculate $S_{t(D)}$

Calculate $C_{t(D)}$

Calculate $T_{t(D)}$

Calculate $D_{t(D)}$

Program Listings

```
101 XEQ 03
102 XEQ 00
103 RCL 08
104 9
105 +
106 RCL 06
107 RCL IND
Y
108 /
109 "SD"
110 XEQ 02
111 XEQ 00
112 1
113 ST+ 09
114 RCL 02
115 RCL 04
116 RCL 18
117 /
118 +
119 XEQ 03
120 "⊢↑"
121 XEQ 00
122 STO 17
123 RCL 08
124 10
125 +
126 RCL 17
127 RCL IND
Y
128 /
129 "SD"
130 XEQ 02
131 "⊢↑"
132 1
133 ST- 09
134 RDN
135 GTO 00
136◆LBL C
137 RCL 14
138 RCL 15
139 RCL 09
140 1
141 -
142 /
143 "MD"
144 XEQ 00
145 /
146 "T"
147 XEQ 00
148 RCL 03
149 "Σe2"
150 XEQ 00
```

Calculate $D_{t(S)}$

Calculate D_{t+1}

Calculate $D_{t+1(S)}$

Calculate MD

Calculate T

Σe_i^2

```
151 "B"
152 RCL 09
153 XEQ 00
154 /
155 "V"
156◆LBL 00
157 "⊢="
158 ARCL X
159 AVIEW
160 RTN
161◆LBL 03
162 "D"
163◆LBL 02
164 FIX 0
165 ARCL 09
166 FIX 2
167 .END.
```

70

80

90

00

V

Output routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Xt	50	
	St_{-1}		
	St		
	Σei^2		
	Tt		
05		55	
	Dt		
	Qi		
	Bi		
10	SV_1	60	
	SV_2		
	SV_3		
	SV_4		
	CFE		
15	CAD	65	
	Temp ei		
	Dt_{+1}		
	α		
	1-α		
20	Xt_{-1} (s)	70	
	Xt_{-1} (D)		
	Xt (D)		
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 023 TOT. REG. 61 USER MODE

ENG ____ FIX 2 SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21		Printer enable	
29			Suppress decimal

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

MONTHLY SEASONAL VARIATION FACTORS BASED ON CENTERED MOVING AVERAGES

Seasonal variation factors are a useful concept in many types of forecasting. There are several methods of developing seasonal moving averages, one of the more common ways being to calculate them as the ratio of the periodic value to a centered moving average for the same period.

For instance, to determine the sales for the 7th month of a given year, a centered moving average for that month would be calculated from sales figures from the 1st thru 12th months of that year and the 1st month of the following year. The seasonal variation factor for the 7th month would then be the ratio of the actual sales in the 7th month to the centered moving average for that month.

Equations:

$$\bar{X}_c = \frac{\frac{X_1}{2} + (X_2 + X_3 + \ldots + X_m) + \frac{X_m + 1}{2}}{m}$$

$$SV = \frac{X_i}{\bar{X}_i}$$

where $\bar{X}_c$ = centered moving average

m = number of elements in the centered moving average

SV = seasonal variation factor

X_i = value of the ith data point

$\bar{X}_i$ = centered moving average of the data point

Example:

Econo-Wise Home Appliance Company's monthly sales for the last 15 months are:

Month	Sales ($K)	Month	Sales ($K)
1	397	9	513
2	376	10	434
3	460	11	562
4	501	12	593
5	455	13	579
6	390	14	601
7	530	15	598
8	560		

Find the centered 12-month moving average and seasonal variation factor for months 7-9.

Keystrokes:	Display
[XEQ] [ALPHA] SIZE [ALPHA] 014	
[XEQ] [ALPHA] SV [ALPHA]	MONTH 1=?
397 [R/S] 376 [R/S]	
460 [R/S] 501 [R/S]	
455 [R/S] 390 [R/S]	
530 [R/S] 560 [R/S]	
513 [R/S] 434 [R/S]	
562 [R/S] 593 [R/S]	MONTH 13=?
579 [R/S]	AVG=488.50 (Centered average for month 7)
[R/S]	SV%=108.50 (Seasonal variation factor)
[R/S]	NEXT MONTH=?
601 [R/S]	AVG=505.46 (Centered average for month 8)
[R/S]	SV%=110.79 (Seasonal variation factor)
[R/S]	NEXT MONTH=?
598 [R/S]	AVG=520.58 (Centered average for month 9)
[R/S]	SV%=98.54 (Seasonal variation factor)

User Instructions

SIZE: 014

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program			
2.	Initialize		[XEQ] SV	MONTH 1=?
3.	Key in 1st month sales	1st	[R/S]	MONTH 2=?
4.	Key in 2nd month sales	2nd	[R/S]	MONTH 3=?
5.	Repeat until month 13 sales are entered	13th		
6.	Calculate moving average for month 7		[R/S]	AVG=
7.	Calculate seasonal variation factor		[R/S]	SV%=
8.	Press [R/S] for next prompt.		[R/S]	NEXT MONTH=?
9.	Key in 14th month sales	14th		
10.	Calculate moving average for month 8.		[R/S]	AVG=
11.	Calculate seasonal variation factor		[R/S]	SV%=
12.	Press [R/S] for next prompt.		[R/S]	NEXT MONTH=?
13.	Repeat steps 9-12 for the balance of the			
	data.			

Program Listings

```
01◆LBL "SV"
02 SF 21
03 12
04 STO 00
05 1
06◆LBL 00
07 FIX 0
08 CF 29
09 "MONTH "
10 ARCL X
11 "⊢=?"
12 PROMPT
13 FIX 2
14 SF 29
15 STO IND
Y
16 RDN
17 1
18 +
19 13
20 X<>Y
21 X<=Y?
22 GTO 00
23◆LBL 01
24 RCL 01
25 2
26 /
27 RCL 02
28 STO 01
29 +
30 RCL 03
31 STO 02
32 +
33 RCL 04
34 STO 03
35 +
36 RCL 05
37 STO 04
38 +
39 RCL 06
40 STO 05
41 +
42 RCL 07
43 STO 06
44 +
45 RCL 08
46 STO 07
47 +
48 RCL 09
49 STO 08
50 +
```

Initialize

Data prompt

Input data

Sum contents of registers and shift

```
51 RCL 10
52 STO 09
53 +
54 RCL 11
55 STO 10
56 +
57 RCL 12
58 STO 11
59 +
60 RCL 13
61 STO 12
62 2
63 /
64 +
65 RCL 00
66 /
67 "AVG="
68 ARCL X
69 AVIEW
70 RCL 06
71 X<>Y
72 /
73 1 E2
74 *
75 "SV%="
76 ARCL X
77 AVIEW
78 "NEXT MO
NTH=?"
79 PROMPT
80 STO 13
81 GTO 01
82 .END.
```

Output

New data

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	n	50	
	X_1		
	X_2		
	X_3		
	X_4		
05	X_5	55	
	X_6		
	X_7		
	X_8		
	X_9		
10	X_{10}	60	
	X_{11}		
	X_{12}		
	X_{13}		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 014 TOT. REG. 32 USER MODE

ENG ___ FIX ___ SCI ___ ON ___ OFF X

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21		Printer enable	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

MULTIPLE LINEAR REGRESSION

This program performs a least squares multiple linear regression for a series of data points x, y, z. Linear regression is a statistical method for finding a straight line that best fits a set of data points. The equation of this straight line expresses the linear relationship between independent (x and y) and dependent (z) variables and is of the form:

$$z = a + bx + cy$$

The three variables are input by pressing [A]. If one or more of the data points was entered incorrectly, simply re-enter the incorrect value(s) and press [///] [A]. Then continue as before. The three coefficients (a, b, c) are calculated by pressing [B].

In addition, the program also calculates the coefficient of determination r^2 ([C]). This is an indication of the "goodness of fit" for the calculated straight line, and is a number between 0 and 1. Values closer to 1 indicate "better" fits than values closer to 0.

Having determined the equation (the [B] key), the user can then project estimates of z for given x, y values ([D]). The sums (Σx_i; Σy_i; Σz_i), the sums of squares (Σx_i^2; Σy_i^2; Σz_i^2), and the sums of cross products ($\Sigma x_i y_i$; $\Sigma x_i z_i$; $\Sigma y_i z_i$) are stored in registers 07-09, 04-06, and 01-03 respectively.

Equations:

$$z = a + bx + cy$$

$$\Sigma z_i = an + b\Sigma x_i + c\Sigma y_i \qquad i = 1, 2, \ldots, n$$

$$\Sigma x_i z_i = a\Sigma x_i + b\Sigma x_i^2 + c\Sigma x_i y_i$$

$$\Sigma y_i z_i = a\Sigma y_i + b\Sigma x_i y_i + c\Sigma y_i^2$$

$$c = \frac{A - B}{[n\Sigma x_i^2 - (\Sigma x_i)^2]\,[n\Sigma y_i^2 - (\Sigma y_i)^2] - [n\Sigma x_i y_i - (\Sigma x_i)\,(\Sigma y_i)]^2}$$

where:

$$A = [n\Sigma x_i^2 - (\Sigma x_i)^2]\,[n\Sigma y_i z_i - (\Sigma y_i)\,(\Sigma z_i)]$$

$$B = [n\Sigma x_i y_i - (\Sigma x_i)\,(\Sigma y_i)]\,[n\Sigma x_i z_i - (\Sigma x_i)\,(\Sigma z_i)]$$

$$b = \frac{[n\Sigma x_i z_i - (\Sigma x_i)\,(\Sigma z_i)] - c[n\Sigma x_i y_i - (\Sigma x_i)\,(\Sigma y_i)]}{n\Sigma x_i^2 - (\Sigma x_i)^2}$$

$$a = \frac{1}{n} (\Sigma z_i - c \Sigma y_i - b \Sigma x_i)$$

$$R^2 = \frac{a\Sigma z_i + b\Sigma x_i z_i + c\Sigma y_i z_i - \frac{1}{n} (\Sigma z_i)^2}{(\Sigma z_i^2) - \frac{(\Sigma z_i)^2}{n}}$$

Example:

A commercial land appraiser has examined 5 vacant lots in the downtown section of a local community, all of which have different depths, frontages, and values as shown below. Based on this data, what is the relationship between depth, frontage, and lot value? What is the coefficient of determination? What predicted value would a lot have with 50 foot depth and 70 foot frontage? With a 75 foot depth and 80 foot frontage?

Lot Depth (feet)	Lot Frontage (feet)	Lot Value
70	70.8	$101,000
90	60.0	82,190
85	90.0	170,000
40	70.0	100,000
100	60.0	90.000

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 015	
[XEQ] [ALPHA] MULT [ALPHA]	0.00
70 [ENTER↑] 70.8 [ENTER↑]	
101000 [A]	N=1.00
90 [ENTER↑] 60 [ENTER↑]	
82190 [A]	N=2.00
85 [ENTER↑] 90 [ENTER↑]	
170000 [A]	N=3.00
40 [ENTER↑] 70 [ENTER↑]	
100000 [A]	N=4.00
100 [ENTER↑] 60 [ENTER↑]	
90000 [A]	N=5.00
[B]	a=-118,499.03
[R/S]	b=314.71
[R/S]	c=2,892.02
[C]	R2=0.98
50 [ENTER↑] 70 [D]	Z=99,678.08
75 [ENTER↑] 80 [D]	Z=136,466.08

User Instructions

SIZE: 015

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program and set USER mode		[USER]	
2.	Initialize		[XEQ] MULT	0.00
3.	Key in x, y and corresponding z value	x	[ENTER↑]	
		y	[ENTER↑]	
		z	[A]	N=
4.	Repeat step 3 for all data			
5.	If data was input incorrectly, re-enter			
	incorrect x, y and z values.	x	[ENTER↑]	
		y	[ENTER↑]	
		z	[///] [A]	N=
6.	Calculate coefficients.		[B]	a=
			[R/S]	b=
			[R/S]	c=
7.	Calculate coefficient of determination(r^2)		[C]	R2=
8.	Key in x and y values and calculate the			
	estimated z value, $\hat{z}$, displayed as			
	Z↑ (repeat as often as desired)	x	[ENTER↑]	
		y	[D]	Z↑=
9.	For a new case, go to step 2.			

Program Listings

```
01◆LBL "MUL
T"
02 FIX 2
03 SF 21
04 CLRG
05 CF 01
06 CLX
07 RTN
08◆LBL A
09 STO 12
10 RDN
11 STO 11
12 RDN
13 STO 10
14 7
15 STO 14
16 RDN
17 XEQ 01
18 8
19 STO 14
20 RCL 11
21 XEQ 01
22 9
23 STO 14
24 RCL 12
25 XEQ 01
26 RCL 10
27 RCL 11
28 *
29 FS? 01
30 CHS
31 ST+ 01
32 RCL 10
33 RCL 12
34 *
35 FS? 01
36 CHS
37 ST+ 02
38 RCL 11
39 RCL 12
40 *
41 FS? 01
42 CHS
43 ST+ 03
44 1
45 FS?C 01
46 CHS
47 ST+ 00
48 RCL 00
49 "N="
50 ARCL X
```

Initialize

Input x_i, y_i, z_i

Compute

Σx_i, Σy_i, Σz_i

$\Sigma x_i^2, \Sigma y_i^2, \Sigma z_i^2$

$\Sigma x_i y_i, \Sigma y_i z_i,$

$\Sigma z_i x_i$

```
51 PROMPT
52◆LBL 01
53 FS? 01
54 CHS
55 ST+ IND
14
56 RCL 14
57 3
58 -
59 STO 14
60 RDN
61 X↑2
62 FS? 01
63 CHS
64 ST+ IND
14
65 RTN
66◆LBL B
67 RCL 00
68 RCL 04
69 *
70 RCL 07
71 X↑2
72 -
73 STO 13
74 RCL 00
75 RCL 03
76 *
77 RCL 08
78 RCL 09
79 *
80 -
81 *
82 STO 12
83 RCL 00
84 RCL 01
85 *
86 RCL 07
87 RCL 08
88 *
89 -
90 STO 10
91 RCL 00
92 RCL 02
93 *
94 RCL 07
95 RCL 09
96 *
97 -
98 STO 11
99 *
```

Subroutine for

Σx_i, ...

Σx_i^2, ...

Calculate a, b, c

Program Listings

```
100 RCL 12
101 X<>Y
102 -
103 RCL 13
104 RCL 00
105 RCL 05
106 *
107 RCL 08
108 X↑2
109 -
110 *
111 RCL 10
112 X↑2
113 -
114 /
115 STO 12
116 RCL 11
117 RCL 10
118 RCL 12
119 *
120 -
121 RCL 13
122 /
123 STO 11
124 RCL 09
125 RCL 12
126 RCL 08
127 *
128 -
129 RCL 11
130 RCL 07
131 *
132 -
133 RCL 00
134 /
135 STO 10
136 "a"
137 XEQ 00
138 RCL 11
139 "b"
140 XEQ 02
141 RCL 12
142 "c"
143 GTO 02
144◆LBL C
145 RCL 10
146 RCL 09
147 *
148 RCL 11
149 RCL 02
150 *
151 +
```

$c \rightarrow R_{12}$

$b \rightarrow R_{11}$

$a \rightarrow R_{10}$

Calculate r^2

```
152 RCL 12
153 RCL 03
154 *
155 +
156 RCL 09
157 X↑2
158 RCL 00
159 /
160 -
161 RCL 06
162 RCL 09
163 X↑2
164 RCL 00
165 /
166 -
167 /
168 "R2"
169 GTO 00
170◆LBL D
171 RCL 12
172 *
173 X<>Y
174 RCL 11
175 *
176 +
177 RCL 10
178 +
179 "Z↑"
180 GTO 02
181◆LBL a
182 SF 01
183 GTO A
184◆LBL 00
185 ADV
186◆LBL 02
187 "⊢="
188 ARCL X
189 AVIEW
190 .END.
```

90	
00	

Calculate $\hat{z}$ for given x, y

Correction of input values

Output routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	n	50	
	$\Sigma x_i y_i$		
	$\Sigma x_i z_i$		
	$\Sigma y_i z_i$		
	Σx_i^2		
05	Σy_i^2	55	
	Σz_i^2		
	Σx_i		
	Σy_i		
	Σz_i		
10	x_{ij} a	60	
	y_{ij} b		
	z_{ij} c		
	Used		
	Used		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 015　TOT. REG. 50　USER MODE

ENG ___　FIX 2　SCI ___　ON X　OFF ___

DEG ___　RAD ___　GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		correction	
21		Printer enable	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

NORMAL, INVERSE NORMAL, t AND F DISTRIBUTIONS

These programs evaluate the standard normal density function f(x), the normal integral Q(x) for a given x and the cumulative distribution P(x) for a given x and degrees of freedom ν, and the integral of the F distribution for given values of x.

Equations:

1. Standard normal density

$$f(x) = \frac{1}{\sqrt{2\pi}} e^{-\frac{x^2}{2}}$$

2. Normal integral

$$Q = \frac{1}{\sqrt{2\pi}} \int_x^\infty e^{-\frac{t^2}{2}}\, dt.$$

(Solving for x given Q as the inverse normal distribution)

3. t density function

$$f(x) = \frac{\Gamma\left(\frac{\nu+1}{2}\right)}{\sqrt{\pi\nu}\,\Gamma\left(\frac{\nu}{2}\right)} \left(1 + \frac{x^2}{\nu}\right)^{-\frac{\nu+1}{2}}$$

4. Cumulative distribution function

$$P(x) = \int_{-\infty}^{x} f(y)dy$$

5. F distribution

$$P(x) = \int_x^\infty \frac{\Gamma\left(\frac{\nu_1+\nu_2}{2}\right) y^{\frac{\nu_1}{2}-1} \left(\frac{\nu_1}{\nu_2}\right)^{\frac{\nu_1}{2}}}{\Gamma\left(\frac{\nu_1}{2}\right)\Gamma\left(\frac{\nu_2}{2}\right)\left(1+\frac{\nu_1}{\nu_2}y\right)^{\frac{\nu_1+\nu_2}{2}}}\, dy$$

The iterative technique used by the program to solve this integral requires that at least one of the degrees of freedom (ν, or ν_2) be even.

Reference:

Abramowitz and Stegun, Handbook of Mathematical Functions, National Bureau of Standards, 1970.

Example 1: (Program 1)

Using normal distribution, find f(x) and Q(x) for x = 1.18.

Keystrokes:	Display:	
[USER]		(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 016		
[XEQ] [ALPHA] NORMAL [ALPHA]	0.0000	
1.18 [A]	F=0.1989	(f(1.18))
1.18 [B]	Q=0.1190	(Q(1.18))

Example 2: (Program 2)

Using t distribution, find f(x) and P(x) for x = 2.2, ν = 11.

Keystrokes:	Display:	
[USER]		(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 013		
[XEQ] [ALPHA] T [ALPHA]	0.0000	
11 [A]	V=11.0000	
2.2 [B]	F=0.0437	(f(2.2))
2.2 [C]	P=0.9750	(P(2.2))

Example 3: (Program 3)

Using F distribution, find P(x) for x = 4.21, ν_1 = 7, ν_2 = 6.

Keystrokes:	Display:	
[USER]		(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 008		
[XEQ] [ALPHA] FDIST [ALPHA]	0.0000	
7 [A]	V1=7.0000	
6 [B]	V2=6.0000	
4.21 [C]	P=0.0499	(P(4.21))

User Instructions

SIZE: 016

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program and set USER mode		[USER]	
2.	Initialize		[XEQ] NORMAL	0.0000
3.	Key in x to compute f(x)	x	[A]	F=
4.	Key in x to compute Q(x)	x	[B]	Q=
5.	Key in Q(x) to compute x	Q(x)	[C]	X=
6.	For a new case, return to step 3,4, or 5.			

Program Listings

```
 01◆LBL "NOR
MAL"
 02 CF 00
 03 CF 01
 04 CF 02
 05 SF 21
 06 .2316419
 07 STO 02
 08 1.330274
429
 09 STO 03
 10 -1.82125
5978
 11 STO 04
 12 1.781477
937
 13 STO 05
 14 -.356563
782
 15 STO 06
 16 .3193815
3
 17 STO 07
 18 2.515517
 19 STO 09
 20 .802853
 21 STO 10
 22 .010328
 23 STO 11
 24 1.432788
 25 STO 12
 26 .189269
 27 STO 13
 28 .001308
 29 STO 14
 30 CLX
 31 RTN
 32◆LBL A
 33 STO 00
 34 X↑2
 35 2
 36 /
 37 CHS
 38 E↑X
 39 PI
 40 2
 41 *
 42 SQRT
 43 /
 44 STO 01
 45 FS? 02
```

Initialize and store coefficients

Input x to calculate f(x)

```
 46 RTN
 47 "F"
 48 GTO 09
 49◆LBL B
 50 STO 00
 51 SF 02
 52 XEQ A
 53 CF 02
 54 RCL 00
 55 X<0?
 56 GTO 01
 57 SF 00
 58◆LBL 13
 59 1
 60 RCL 00
 61 RCL 02
 62 *
 63 +
 64 1/X
 65 ENTER↑
 66 ENTER↑
 67 ENTER↑
 68 RCL 03
 69 *
 70 RCL 04
 71 +
 72 *
 73 RCL 05
 74 +
 75 *
 76 RCL 06
 77 +
 78 *
 79 RCL 07
 80 +
 81 *
 82 RCL 01
 83 *
 84 "Q"
 85 FS? 00
 86 GTO 09
 87 RTN
 88◆LBL 01
 89 CF 00
 90 RCL 00
 91 CHS
 92 STO 00
 93 XEQ 13
 94 1
 95 X<>Y
 96 -
```

Input x to calculate Q(x)

Program Listings

```
 97 STO 08
 98 "Q"
 99 GTO 09
100◆LBL C
101 X<0?
102 GTO 00
103 1
104 X<=Y?
105 GTO 00
106 RDN
107 .5
108 X<>Y
109 X>Y?
110 XEQ 08
111 X↑2
112 1/X
113 LN
114 SQRT
115 STO 15
116 RCL 11
117 *
118 RCL 10
119 +
120 RCL 15
121 *
122 RCL 09
123 +
124 RCL 15
125 RCL 14
126 *
127 RCL 13
128 +
129 RCL 15
130 *
131 RCL 12
132 +
133 RCL 15
134 *
135 1
136 +
137 /
138 RCL 15
139 X<>Y
140 -
141 FS? 01
142 CHS
143 GTO 10
144◆LBL 08
145 SF 01
146 1
147 -
```

Input Q(x) to calculate x

For Q(x) < .5

```
148 CHS
149 RTN
150◆LBL 00
151 0
152 /
153◆LBL 10
154 "X"
155◆LBL 09
156 CF 02
157 CF 01
158 CF 00
159 "⊢="
160 ARCL X
161 AVIEW
162 .END.
```

Data error

Output routine

70

80

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	x	50	
	f(x)		
	r		
	b_5		
	b_4		
05	b_3	55	
	b_2		
	b_1		
	USED		
	c_0		
10	c_1	60	
	c_2		
	d_1		
	d_2		
	d_3		
15	t	65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 016 TOT. REG. 59 USER MODE
ENG ____ FIX ____ SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Positive	
01		Q(x) < .5	
02		Q(x)	
21		Printer enable	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

User Instructions

SIZE: 013

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program and set USER mode		[USER]	
2.	Initialize		[XEQ] T	0.0000
3.	Key in degrees of freedom ν	ν	[A]	V=
4.	Key in x to compute f(x)	x	[B]	F=
5.	Key in x to compute P(x)	x	[C]	P=
6.	For a new case with the same ν, go to			
	step 4 or 5. For a new case with a			
	different ν, go to step 3.			

Program Listings

```
01◆LBL "T"
02 SF 21
03 CF 01
04 CLX
05 RTN
06◆LBL A
07 STO 00
08 "V"
09 GTO 10
10◆LBL B
11 STO 12
12 RCL 00
13 XEQ 11
14 STO 10
15 RCL 00
16 1
17 +
18 XEQ 11
19 STO 09
20 RCL 12
21 RCL 09
22 RCL 10
23 /
24 PI
25 RCL 00
26 *
27 SQRT
28 /
29 1
30 RCL 12
31 X↑2
32 RCL 00
33 /
34 +
35 RCL 00
36 1
37 +
38 2
39 /
40 CHS
41 Y↑X
42 *
43 "F"
44 GTO 10
45◆LBL 11
46 1
47 STO 03
48 X<>Y
49 2
50 /
51 STO 01
```

Initialize

Input ν

Input x to calculate f(x)

Calculate Γ(ν/2)

```
52 INT
53 LASTX
54 X≠Y?
55 GTO 01
56 1
57 -
58 FACT
59 STO 03
60 RTN
61◆LBL 01
62 .5
63 X=Y?
64 GTO 02
65 X<>Y
66 1
67 -
68 ST* 03
69 GTO 01
70◆LBL 02
71 PI
72 SQRT
73 RCL 03
74 *
75 STO 03
76 RTN
77◆LBL C
78 STO 12
79 ABS
80 RCL 00
81 RAD
82 SQRT
83 /
84 ATAN
85 STO 02
86 RCL 00
87 2
88 /
89 INT
90 LASTX
91 X≠Y?
92 GTO 04
93 0
94 STO 05
95◆LBL 12
96 RCL 02
97 COS
98 X↑2
99 STO 03
100 RCL 02
101 SIN
102 STO 04
```

Check for ν/2 = 1/2

Γ(½)

Input x to calculate P(x)

Program Listings

```
103 RCL 00
104 2
105 X=Y?
106 GTO 08
107 /
108 INT
109 STO 11
110 1
111 STO 06
112 DSE 11
113 GTO 03
114 GTO 13
115◆LBL 03
116 RCL 03
117 *                    For even ν
118 RCL 05
119 1
120 +
121 *
122 LASTX
123 1
124 +
125 STO 05
126 /
127 ST+ 06
128 DSE 11
129 GTO 03
130 RCL 06
131◆LBL 13
132 RCL 04
133 *
134 FS? 01
135 RTN
136 RCL 12
137 GTO 06
138◆LBL 04
139 RCL 02
140 2
141 *                    For odd ν
142 PI
143 /
144 STO 07
145 RCL 00
146 1
147 STO 05
148 X=Y?
149 GTO 09
150 SF 01
151 XEQ 12
152 CF 01
153 RCL 02
154 COS
155 *
156 2
157 *
158 PI
159 /
160 RCL 07
161 +
162 RCL 12
163 GTO 06
164◆LBL 09                For odd ν
165 RCL 07
166 RCL 12
167 GTO 06
168◆LBL 08
169 RCL 04                For even ν
170 RCL 12
171◆LBL 06
172 X>0?                  Calculate P(x)
173 GTO 00                from R(x)
174 X<>Y
175 1                     for x ≤ 0
176 -
177 CHS
178 2
179 /
180 "P"
181 GTO 10
182◆LBL 00                Calculate P(x)
183 X<>Y                  for x > 0
184 1
185 +
186 2
187 /
188 "P"
189◆LBL 10
190 "⊢="                  Output routine
191 ARCL X
192 AVIEW
193 .END.
```

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	ν	50	
	USED		
	Θ		
	USED		
	SIN Θ		
05	USED	55	
	USED		
	2Θ/π, R		
	USED		
	Γ(ν+1/2)		
10	Γ(ν/2)	60	
	Index		
	x		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 013 TOT. REG. 47 USER MODE

ENG ____ FIX ____ SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		odd ν	
21		Printer enable	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program and set USER mode.		[USER]	
2.	Initialize		[XEQ] FDIST	0.0000
3.	Key in ν_1	ν_1	[A]	V1=
4.	Key in ν_2	ν_2	[B]	V2=
5.	Key in x to calculate P(x)		[C]	P=
6.	For a new case to to step 3.			

Program Listings

```
01◆LBL "FDI
ST"
02 SF 21
03 CF 01
04 0
05 STO 05
06 RTN
07◆LBL A
08 STO 01
09 "V1"
10 GTO 10
11◆LBL B
12 STO 02
13 "V2"
14 GTO 10
15◆LBL C
16 "P"
17 STO 06
18 RCL 01
19 *
20 RCL 02
21 +
22 LASTX
23 X<>Y
24 /
25 STO 03
26 RCL 01
27 2
28 /
29 FRC
30 X≠0?
31 GTO 01
32◆LBL 00
33 RCL 03
34 RCL 02
35 2
36 STO 07
37 /
38 Y↑X
39 STO 04
40 RCL 01
41 2
42 -
43 2
44 /
45 STO 00
46 X≠0?
47 GTO 05
48 RCL 04
49 GTO 10
50◆LBL 05
```

Initialize

Input ν_1

Input ν_2

If ν_1 odd then go to LBL 01

ν_1 even

```
51 1
52 STO 05
53 RCL 03
54 -
55 STO 03
56 RCL 02
57 2
58 /
59 *
60 ST+ 05
61 DSE 00
62 GTO 03
63 GTO 02
64◆LBL 03
65 RCL 02
66 2
67 +
68 STO 02
69 RCL 07
70 2
71 +
72 STO 07
73 /
74 RCL 03
75 *
76 *
77 ST+ 05
78 DSE 00
79 GTO 03
80◆LBL 02
81 RCL 05
82 RCL 04
83 *
84 RTN
85◆LBL 01
86 RCL 01
87 X<> 02
88 STO 01
89 1
90 RCL 03
91 -
92 STO 03
93 XEQ 00
94 1
95 X<>Y
96 -
97◆LBL 10
98 "⊢="
99 ARCL X
100 AVIEW
101 .END.
```

ν_1 odd

Output routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	USED	50	
	ν_1 or ν_2		
	ν_2 or ν_1		
	t, 1-t		
	USED		
05	USED	55	
	x		
	USED		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 28 USER MODE
ENG ____ FIX ____ SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21		Printer enable	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BASIC STATISTICS FOR TWO VARIABLES

This program calculates means, standard deviations, covariance, correlation coefficient, and coefficients of variation from ungrouped data points $[(x_i,y_i), i=1,2,\ldots,n]$ or grouped data points $[(x_i,y_i,f_i), i=1,2,\ldots,n]$. f_i denotes the frequency of repetition of (x_i,y_i).

Equations:

Mean $$\bar{x} = \frac{1}{n}\sum_{i=1}^{n} x_i \qquad \bar{y} = \frac{1}{n}\sum_{i=1}^{n} y_i$$

Standard deviation

$$s_x = \sqrt{\frac{\Sigma x_i^2 - n\bar{x}^2}{n-1}}$$

$$s_x' = \sqrt{\frac{\Sigma x_i^2 - n\bar{x}^2}{n}}$$

$$s_y = \sqrt{\frac{\Sigma y_i^2 - n\bar{y}^2}{n-1}}$$

$$s_y' = \sqrt{\frac{\Sigma y_i^2 - n\bar{y}^2}{n}}$$

covariance

$$s_{xy} = \frac{1}{n-1}\left(\Sigma x_i y_i - \frac{1}{n}\Sigma x_i y_i\right)$$

$$s_{xy}' = \frac{1}{n}\left(\Sigma x_i y_i - \frac{1}{n}\Sigma x_i y_i\right)$$

correlation coefficient

$$\gamma_{xy} = \frac{s_{xy}}{s_x s_y}$$

coefficients of variation

$$V_x = \frac{s_x}{\bar{x}}\ .100, \qquad V_y = \frac{s_y}{\bar{y}}\ .100$$

Note: n is a positive integer > 1.

Example:

For the following set of data, find the means, standard deviations, covariance, correlation coefficient and coefficients of variation.

x_i	4.8	5.2	4.1	3.8
y_i	15.1	11.5	13.6	14.3
f_i	1	3	2	1

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 017	
[XEQ] [ALPHA] STAT [ALPHA]	0.0000
4.8 [ENTER↑] 15.1 [ENTER↑] 1 [B]	1.0000
5.2 [ENTER↑] 11.5 [ENTER↑] 3 [B]	4.0000
4.1 [ENTER↑] 13.6 [ENTER↑] 2 [B]	6.0000
3.8 [ENTER↑] 14.3 [ENTER↑] 1 [B]	7.0000
[C]	MEANX=4.6286
[R/S]	MEANY=13.0143
[R/S]	VX=13.2429
[R/S]	VY=11.5550
[D]	SX=0.6130
[R/S]	SY=1.5038
[R/S]	SX.=0.5675
[R/S]	SY.=1.3923
[E]	SXY=-0.6538
[R/S]	SXY.=-0.5604
[R/S]	VXY=-0.7093

User Instructions

SIZE: 017

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program and set USER mode		[USER]	
2.	Initialize		[XEQ] STAT	0.0000
3.	For grouped data points, go to step 6.			
	For ungrouped data points, go to			
	step 4.			
4.	For i = 1, 2, ..., n, key in x_i and y_i	x_i	[ENTER↑]	
		y_i	[A]	i
5.	If you made a mistake in keying in x_k			
	and y_k, the correct by	x_k	[ENTER↑]	
		y_k	[///] [A]	i-1
6.	For i = 1, 2, ..., n, key in x_i, y_i and			
	f_i (grouped data)	x_i	[ENTER↑]	
		y_i	[ENTER↑]	
		f_i	[B]	f_i
7.	If you made a mistake in keying in x_k, y_k			
	and f_k, then correct by	x_k	[ENTER↑]	
		y_k	[ENTER↑]	
		f_k	[///] [B]	$f_i - f_k$
8.	Calculate means ($\bar{x}$ and $\bar{y}$) and coefficients			
	of variation (V_x and V_y)		[C]	MEANX =
			[R/S]	MEANY =
			[R/S]	VX =
			[R/S]	VY =
9.	Calculate standard deviations			
	(s_x, s_y, s_x' and s_y')		[D]	SX =
			[R/S]	SY =
			[R/S]	SX. =

User Instructions

SIZE: 017

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
			[R/S]	SY.=
10.	Calculate covariance and correlation			
	coefficient (s_{xy}, s_{xy}', and V_{xy})		[E]	SXY=
			[R/S]	SXY.=
			[R/S]	VXY=
11.	For a new case, go to step 2.			

Program Listings

```
 01◆LBL "STA
T"
 02 CLRG
 03 SF 21
 04 CF 00
 05 CF 01
 06 CLX
 07 RTN
 08◆LBL a
 09 X<>Y
 10 Σ-
 11 RTN
 12◆LBL A
 13 X<>Y
 14 Σ+
 15 RTN
 16◆LBL B
 17 STO 05
 18 FS? 01
 19 CHS
 20 ST+ 01
 21 RDN
 22 STO 04
 23 RDN
 24 STO 03
 25 R↑
 26 R↑
 27 ABS
 28 STO 06
 29◆LBL 02
 30 RCL 04
 31 RCL 03
 32 XEQ 03
 33 DSE 06
 34 GTO 02
 35 RCL 01
 36 STO 16
 37 RTN
 38◆LBL b
 39 SF 01
 40 XEQ B
 41 CF 01
 42 RTN
 43◆LBL 03
 44 FS? 01
 45 GTO 04
 46 Σ+
 47 RTN
 48◆LBL 04
 49 Σ-
 50 RTN
```

Initialize

Correction for x_k, y_k

Input x_i, y_i

Input x_i, y_i, f_i

Correction for x_k, y_k, f_k

```
 51◆LBL C
 52 MEAN
 53 STO 00
 54 "MEANX="
 55 XEQ 00
 56 X<>Y
 57 STO 02
 58 "MEANY="
 59 XEQ 00
 60 SDEV
 61 1 E2
 62 *
 63 X<>Y
 64 LASTX
 65 *
 66 X<>Y
 67 RCL 00
 68 /
 69 "VX="
 70 XEQ 00
 71 X<>Y
 72 RCL 02
 73 /
 74 "VY="
 75 XEQ 00
 76 RTN
 77◆LBL D
 78 SDEV
 79 "SX="
 80 XEQ 00
 81 X<>Y
 82 "SY="
 83 XEQ 00
 84 X<>Y
 85◆LBL 01
 86 RCL 16
 87 ENTER↑
 88 X<>Y
 89 1
 90 -
 91 /
 92 SQRT
 93 /
 94 FS?C 00
 95 GTO 05
 96 "SX.="
 97 XEQ 00
 98 LASTX
 99 SDEV
100 X<>Y
101 SF 00
```

Calculate $\bar{x}$, $\bar{y}$, V_x, V_y

Calculate S_x, S_y, S_x', S_y'

Program Listings

```
102 GTO 01
103◆LBL 05
104 "SY.="
105 GTO 00
106◆LBL E
107 MEAN
108 X<>Y
109 STO 02
110 RCL 15
111 RCL 11
112 RCL 02
113 *
114 -
115 RCL 16
116 1
117 -
118 /
119 STO 07
120 "SXY="
121 XEQ 00
122 RCL 16
123 ENTER↑
124 X<>Y
125 1
126 -
127 /
128 /
129 "SXY.="
130 XEQ 00
131 SDEV
132 RCL 07
133 /
134 *
135 1/X
136 "VXY="
137◆LBL 00
138 ARCL X
139 AVIEW
140 .END.
```

Calculate S_{xy}, S_{xy}'

Output routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	$\bar{x}$	50	
	Σfi		
	y		
	x_i		
	y_i		
05	f_i	55	
	USED		
	USED		
10		60	
	Σx_i		
	Σx_i^2		
	Σy_i		
	Σy_i^2		
15	$\Sigma x_i y_i$	65	
	n		
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 017 TOT. REG. 52 USER MODE
ENG ___ FIX ___ SCI ___ ON X OFF ___
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		S_y'	
01		Correction	
21		Printer enable	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

MOVING AVERAGE

In a moving average, a specified number of data points are averaged. When there is a new piece of input data, the oldest piece of data is discarded to make room for the latest input. This replacement scheme makes the moving average a valuable tool in following trends. The fewer the number of data points, the more trend sensitive the average becomes. With a large number of data points, the average behaves more like a regular average, responding slowly to new input data.

This program allows for a moving average of 14 points with no memory modules and 64 more points with each additional memory module. It also allows for more than one moving average to be computed with the same set of data. For example, instead of obtaining only a 6 month moving average, you could obtain a 3 month and 12 month as well with the same data. For each additional moving average you wish to compute, the maximum number of data points is reduced by three. For example if you have no memory modules and wish to compute 2 moving averages with the same data, the maximum size for an average is 11.

This program is most useful when a card reader is available. With data card(s) to remember your old data points you need only input the most recent data point to compute current moving average(s).

Example 1:

A six period and three period moving average is desired to project monthly sales. The first 6 months of sales follows:

Month	1	2	3	4	5	6
Sales	125	183	207	222	198	240

Compute the 3 month moving average for months 3,4,5, and 6, and the 6 month moving average for month 6.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 006	
[XEQ] [ALPHA] AVG [ALPHA]	NO. OF AVGS ?
2 [R/S]	N MAX?
6 [R/S]	SET SIZE 15 (mark a data card SIZE 015)
[XEQ] [ALPHA] SIZE [ALPHA] 015 [R/S]	N2?
3 [R/S]	DATA ?
125 [R/S]	DATA ?
183 [R/S]	DATA ?
207 [R/S]	DATA ?
[R/S]	MA3=171.6667
[R/S] [R/S]	DATA ?
222 [R/S]	DATA ?
[R/S]	MA3=204.0000
[R/S] [R/S]	DATA ?
198 [R/S]	DATA ?
[R/S]	MA3=209.0000
[R/S] [R/S]	DATA ?
240 [R/S]	DATA ?
[R/S]	MA3=220.0000
[R/S]	MA6=195.8333

Now record the data for example 2.

[XEQ] [ALPHA] UPDATE [ALPHA]	RDY 01 OF 01

Insert one side of the data card into the card reader.

Now turn the calculator off assume a month has passed. Turn the calculator on and load the program.

Example 2:

The actual sales for the seventh month totaled 225 units. Compute new moving averages and output the current points in the averages.

Keystrokes:	Display	
[XEQ] [ALPHA] SIZE [ALPHA] 015		(as marked on the data card)
load the data card		
[XEQ] [ALPHA] PT [ALPHA]	DATA ?	
225 [R/S]	DATA ?	
[R/S]	MA3=221.0000	
[R/S]	MA6=212.5000	
[XEQ] [ALPHA] OUT [ALPHA]	PT1=225.0000	
[R/S]	PT2=240.0000	
[R/S]	PT3=198.0000	
[R/S]	PT4=222.0000	
[R/S]	PT5=207.0000	
[R/S]	PT6=183.0000	

User Instructions

SIZE: 6+

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program.			
2.	If starting new moving average(s) go to			
	step 4.			
3.	Set size to that marked on the data card(s)			
	(during step 7) containing previous moving			
	average(s) data load the card(s) and go			
	to step 11.		[XEQ] SIZE nnn	
4.	Initialize		[XEQ] AVG	NO. OF AVGS ?
5.	Key in the number of moving averages you			
	wish to compute with the one set of data.	k	[R/S]	N MAX?
6.	Key in the number of points that the			
	longest moving average will deal with.	n_{max}	[R/S]	SET SIZE nnn
7.	Set the required size and mark your data			
	cards with the size (for step 3.)		[XEQ] SIZE nnn	
			[R/S]	N2?(or)DATA?
8.	Key in the length of the other moving			
	averages as they are asked for. When			
	"DATA?" is displayed continue to step 9.	n_i	[R/S]	N(i+1)?orDATA?
9.	Key in data as desired. When all the			
	points desired are in, go to step 11 or			
	press [R/S] and continue to step 10.	data	[R/S]	DATA ?
			or	
			[R/S]	MA(i)=
10.	Obtain the other moving averages by			
	pressing [R/S] for each. When one more			
	[R/S] is pressed than is needed you are			
	back to the data input routine,			

User Instructions

SIZE: 6+

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	go to step 9.		[R/S]	MA(i)=
				(or) DATA ?
11.	Any of the following steps can be done in			
	any order and at any time after step 8.			
12.	To input more data:		[XEQ]PT	DATA ?
	then go to step 9.			
13.	To output data in newest to oldest format:		[XEQ] OUT	PT1=
			[R/S]	PT2=
			⋮	⋮
14.	To obtain current moving averages:		[XEQ]AVGS	MA(i)=
	then go to step 10.			
15.	To store current data on cards:		[XEQ]UPDATE	RDY () OF ()
	then load cards requested			

Program Listings

```
 01*LBL "AVG
"
 02 CLRG
 03 "NO. OF
AVGS ?"
 04 PROMPT
 05 1
 06 +
 07 STO 00
 08 3
 09 *
 10 STO 01
 11 "N MAX?"
 12 PROMPT
 13 STO 05
 14 RCL 01
 15 +
 16 FIX 0
 17 CF 29
 18 "SET SIZ
E "
 19 ARCL X
 20 PROMPT
 21 RCL 00
 22 3
 23 *
 24 .96
 25 -
 26 1 E3
 27 /
 28 4
 29 +
 30 STO 02
 31 2
 32 GTO 04
 33*LBL 03
 34 "N"
 35 ARCL X
 36 "⊢?"
 37 CF 22
 38 PROMPT
 39 FC?C 22
 40 GTO "PT"
 41 STO IND
02
 42 CHS
 43 1
 44 ST- 02
 45 RDN
 46 STO IND
02
```

Initialization

n_{max}

Prompt for correct size

n_i

Pointer:

```
 47 RDN
 48 1
 49 +
 50*LBL 04
 51 ISG 02
 52 GTO 03
 53*LBL "PT"
 54 CF 22
 55 "DATA ?"
 56 PROMPT
 57 RCL 01
 58 FC?C 22
 59 GTO "AVG
S"
 60 1 E3
 61 /
 62 4.00003
 63 +
 64 STO 02
 65 1
 66 -
 67 X<>Y
 68*LBL 01
 69 ST+ IND
Y
 70 RCL IND
02
 71 RCL 05
 72 MOD
 73 RCL 01
 74 +
 75 RCL IND
X
 76 ST- IND
T
 77 ISG IND
02
 78*LBL 00
 79 RDN
 80 RDN
 81 ISG Y
 82 ISG 02
 83 GTO 01
 84 RCL 04
 85 1
 86 -
 87 RCL 05
 88 MOD
 89 RCL 01
 90 +
 91 X<>Y
```

Input a data point

If no point calculate average

Add pt. to Σ's

Subtract old pt.

Point to next oldest pt.

Program Listings

```
 92 STO IND
Y
 93 GTO "PT"
 94◆LBL "AVG
S"
 95 CF 29
 96 RCL 01
 97 3
 98 -
 99 .00205
100 +
101 STO 02
102◆LBL 02
103 RCL IND
02
104 2
105 ST+ 02
106 RDN
107 RCL IND
02
108 /
109 "MA"
110 FIX 0
111 ARCL IND
 02
112 "⊢="
113 FIX 4
114 ARCL X
115 PROMPT
116 DSE 02
117 GTO 02
118 GTO "PT"
119◆LBL "OUT
"
120 RCL 05
121 1 E3
122 /
123 1
124 +
125 STO 02
126 RCL 04
127◆LBL 05
128 1
129 -
130 ENTER↑
131 ENTER↑
132 RCL 05
133 MOD
134 RCL 01
135 +
136 "PT"
```

Compute the moving averages

Store counter

Recall Σ_i

Recall n_i

Output moving avg.

Output points

Store counter

Scale pointer

```
137 FIX 0
138 ARCL 02
139 "⊢="
140 FIX 4
141 ARCL IND
 X
142 PROMPT
143 RDN
144 ISG 02
145 GTO 05
146 GTO "OUT
"
147◆LBL "UPD
ATE"
148 WDTA
149 RTN
150 .END.
```

Output point

Update data card(s)

80	
90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	k + 1	50	
	data array pointer		
	counter		
	Σ_1		
	pointer		
05	n_1 is also n_{max}	55	
	Σ_2		
	pointers		
	n_2		
	:		
10	data	60	
	data		
	data		
	:		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 6+ TOT. REG. 49+ USER MODE
ENG ___ FIX ___ SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
22		data	no data
29		decimal	no decimal

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

GOMPERTZ CURVE TREND ANALYSIS

A useful curve for evaluating sales trends, etc., is the Gompertz curve. This is a "growth" curve having a general "S" shape and may be used to describe series of data where the early rate of growth is small, then accelerates for a period of time and then slows again as the time grows long. The sales curves for many products follow this trend during the introductory, growth and early mature phases.

The data points to be fit to a Gompertz curve should be equally spaced along the X (or time) axis and all the data points must be positive. The points are divided serially into 3 groups for data entry.

This program processes the data, fits it to a Gompertz curve and calculates extimated values for future data points. The 3 constants (a,b,c) which characterize the curve are available if desired.

Equations:

$$y=ca^{b^x} \qquad \text{where a,b,c,x, and y are positive}$$

$$b = \left(\frac{s_3 - s_2}{s_2 - s_1}\right)^{1/n}$$

$$c = \exp\left[\frac{1}{n}\left(\frac{s_1 s_3 - s_2^2}{s_1 + s_3 - 2s_2}\right)\right]$$

$$a = \exp\left[\frac{(b-1)\ (s_2 - s_1)}{b(b^n-1)^2}\right]$$

where s_1, s_2, and s_3 are:

$$s_1 = \sum_{i=1}^{n} \ln y_i = n\ln c + \ln a \frac{b^n-1}{b-1}$$

$$s_2 = \sum_{i=n+1}^{2n} \ln y_i = n\ln c + b^{n+1} \ln a \frac{b^n-1}{b-1}$$

$$s_3 = \sum_{i=2n+1}^{3n} \ln y_i = n\ln c + b^{2n+1} \ln a \frac{b^n-1}{b-1}$$

Example:

The X-Presso Company marketed a revolutionary new coffee brewing maching in 1968. Sales grew at a steady pace for several years then began to slow. The sales records for the first 9 years of the product's life were as follows:

Year	Sales ($K)	Year	Sales ($K)
1	18	6	260
2	41	7	282
3	49	8	322
4	151	9	340
5	188		

What are the projected sales volumes for this product in its 10th and 12th year? What is the maximum yearly sales volume for this product if the present trend continues? What annual sales rate would the curve have predicted for the 5th year of the product's life? (Arrange the data as follows)

Group I	Group II	Group III
18	151	282
41	188	322
49	260	340

Keystrokes:	Display:	
[USER]		(Set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 007		
[XEQ] [ALPHA] GOMP [ALPHA]	0.0000	
18 [A] 41 [A] 49 [A]	3.0000	
151 [B] 188 [B] 260 [B]	6.0000	
282 [C] 322 [C] 340 [C]	9.0000	(Total # of entries)
[D]	a=0.0042	(a)
[R/S]	b=0.6456	(b)
[R/S]	c=373.9220	(c)
10 [E]	Y.=349.0896	($K sales in 10th year)
12 [E]	Y.=363.3649	(#K sales in 12th year)
100 [E]	Y.=373.9220	(Maximum annual sales after long product life)
5 [E]	Y.=202.5965	($K sales in 5th year- actual sales were $188K)

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program and set USER mode		[USER]	
2.	Initialize		[XEQ] COMP	0.0000
3.	Key in the first group of data.	Group 1	[A]	#of entries
4.	Key in the second group of data.	Group 2	[B]	#of entries
5.	Key in the third group of data.	Group 3	[C]	#of entries
6.	Compute coefficients.		[D]	a=
			[R/S]	b=
			[R/S]	c=
7.	Compute estimated value y	x	[E]	y.=

Program Listings

```
01♦LBL "GOM
P"
02 0
03 STO 00
04 STO 01
05 STO 02
06 STO 03
07 SF 21
08 RTN
09♦LBL A
10 LN
11 ST+ 01
12 GTO 00
13♦LBL B
14 LN
15 ST+ 02
16 GTO 00
17♦LBL C
18 LN
19 ST+ 03
20♦LBL 00
21 1
22 ST+ 00
23 RCL 00
24 RTN
25♦LBL D
26 3
27 ST/ 00
28 RCL 03
29 RCL 02
30 -
31 RCL 02
32 RCL 01
33 -
34 /
35 RCL 00
36 1/X
37 Y↑X
38 STO 05
39 RCL 01
40 RCL 03
41 *
42 RCL 02
43 X↑2
44 -
45 RCL 01
46 RCL 03
47 +
48 RCL 02
49 2
50 *
```

Initialize

Accumulate variables

Compute a,b,c

```
51 -
52 /
53 RCL 00
54 /
55 E↑X
56 STO 06
57 RCL 05
58 1
59 -
60 RCL 05
61 RCL 00
62 Y↑X
63 1
64 -
65 X↑2
66 /
67 RCL 05
68 /
69 RCL 02
70 RCL 01
71 -
72 *
73 E↑X
74 STO 04
75 "a"
76 XEQ 01
77 "b"
78 RCL 05
79 XEQ 01
80 "c"
81 RCL 06
82♦LBL 01
83 "⊢="
84 ARCL X
85 AVIEW
86 RTN
87♦LBL E
88 RCL 05
89 X<>Y
90 Y↑X
91 RCL 04
92 X<>Y
93 Y↑X
94 RCL 06
95 *
96 "Y."
97 XEQ 01
98 .END.
```

Output routine

Compute $\hat{y}$

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	n	50	
	s_1		
	s_2		
	s_3		
	a		
05	b	55	
	c		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 26 USER MODE

ENG ___ FIX ___ SCI ___ ON X OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21		Printer enable	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BREAK-EVEN ANALYSIS

Break-even analysis is a technique for analyzing the relationships among fixed costs, variable costs, and income. Until the break-even point is reached, at the intersection of the total income and total cost lines, the producer operates at a loss. After the break-even point, each unit produced and sold makes a profit. Break-even analysis may be represented as follows:



Given four of the following variables: fixed costs (F), sales price per unit (P), variable costs per unit (V), number of units sold (U), and gross profit (GP), this program evaluates the remaining variable. To calculate the break-even values, simply let the gross profit equal zero.

The degree of operating leverage (OL) at a point is defined as the ratio of the percentage change in net operating income to the percentage change in units sold. The greatest degree of operating leverage is found near the break-even point, where a small change in sales may produce a very large increase in profits. This happens because the profits are close to zero near the break-even point. Likewise, firms with a small degree of operating leverage are operating farther from the break-even point, and they are relatively insensitive to changes in sales volume.

The necessary inputs to calculate the degree of operating leverage are fixed costs (F), sales price per unit (P), variable costs per unit (V), and number of units (U).

For subsequent calculations, it is necessary only to input <u>new</u> data.

Break Even Analysis

$$GP = U(P-V) - F$$

$$OL = \frac{U(P-V)}{U(P-V) - F}$$

Example 1: The Cooper Company sells finance textbooks at $13 each. Given costs and revenues below, how many textbooks must be sold to break even?

FIXED COSTS	
Typesetting	$ 4,000
Graphics production	5,000
Printing and binding	3,000
Total fixed costs	$12,000
VARIABLE COSTS PER COPY	
Distribution	$ 1.00
Commissions	3.75
Royalties	2.00
Total variable costs per copy	$ 6.75
Sales price per copy	$13.00

Example 2: What is the Cooper Company's degree of operating leverage at 2000 units? At 5000 units?

Keystrokes:	Display:
[USER]	(Set USER mode.)
[XEQ] [ALPHA] SIZE [ALPHA] 007	
[XEQ] [ALPHA] BEA [ALPHA]	FIXED ?
12000 [R/S]	PRICE ?
13 [R/S]	VARIABLE ?
6.75 [R/S]	UNITS ?
[R/S]	G. PROFIT ?
0 [R/S]	UNITS=1920.00
[B]	UNITS ?
2000 [R/S]	% LEV.=25.00
[B]	UNITS ?
5000 [R/S]	% LEV.=1.62

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Key in the program and set USER mode		[USER]	
2	Initialize		[XEQ] BEA	FIXED ?
3	Input 4 of the following: fixed cost;	F	[R/S]	PRICE ?
	price;	P	[R/S]	VARIABLE ?
	variable cost;	V	[R/S]	UNITS ?
	no. of units;	U	[R/S]	G. PROFIT ?
	and gross profit.	G.P.	[R/S]	FIXED=$( )
3	When prompted for the unknown quantity,			-or- PRICE=$( )
	press [R/S] (make no input). The			-or- VAR.=$( )
	unknown will be calculated automatically			-or- UNITS=( )
	when all the data is input			-or- G.P.=( )
4	To find percent operating leverage		[B]	UNITS ?
5	Input number of units	U	[R/S]	% LEV.=( )

Program Listings

```
 01◆LBL "BEA
"
 02 1.1
 03 STO 00
 04 CF 22
 05 "FIXED ?
"
 06 XEQ 09
 07 "PRICE ?
"
 08 XEQ 09
 09 "VARIABL
E ?"
 10 XEQ 09
 11 "UNITS ?
"
 12 XEQ 09
 13 "G. PROF
IT ?"
 14 XEQ 09
 15 GTO IND
06
 16◆LBL 01
 17 RCL 02
 18 RCL 03
 19 -
 20 RCL 04
 21 *
 22 RCL 05
 23 -
 24 STO 01
 25 "FIXED"
 26 XEQ 00
 27◆LBL 02
 28 RCL 01
 29 RCL 05
 30 +
 31 RCL 04
 32 /
 33 RCL 03
 34 +
 35 STO 02
 36 "PRICE"
 37 XEQ 00
 38◆LBL 03
 39 RCL 02
 40 RCL 01
 41 RCL 05
 42 +
 43 RCL 04
 44 /
```

Initialize and input data

Solve for F

Solve for P

Solve for V

```
 45 -
 46 STO 03
 47 "VAR."
 48 XEQ 00
 49◆LBL 04
 50 RCL 01
 51 RCL 05
 52 +
 53 RCL 02
 54 RCL 03
 55 -
 56 /
 57 STO 04
 58 "UNITS="
 59 ARCL X
 60 PROMPT
 61◆LBL 05
 62 RCL 02
 63 RCL 03
 64 -
 65 RCL 04
 66 *
 67 RCL 01
 68 -
 69 STO 05
 70 "G.P."
 71 XEQ 00
 72◆LBL B
 73 "UNITS ?
"
 74 PROMPT
 75 RCL 02
 76 RCL 03
 77 -
 78 *
 79 STO 06
 80 RCL 06
 81 RCL 01
 82 -
 83 /
 84 "% LEV.=
"
 85 ARCL X
 86 PROMPT
 87◆LBL 00
 88 "⊢=$"
 89 ARCL X
 90 PROMPT
 91 RTN
 92◆LBL 09
 93 PROMPT
```

Solve for U

Solve for G.P.

Solve for OL

Display routine

Input storage routine

Program Listings

```
  94 STO IND
00
  95 RCL 00
  96 FC?C 22
  97 STO 06
  98 ISG 00
  99 RTN
 100 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	pointer	50	
	F		
	P		
	V		
	U		
05	G.P.	55	
	subroutine pointer		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 43 USER MODE

ENG ____ FIX 2 SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
22	C	refer to owner's manual	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

EXPERIENCE (LEARNING) CURVE FOR MANUFACTURING COST

Many production process costs vary with output in close relation to the learning curve:

$$C_n = C_1 n^{\log r/\log 2}$$

where C_1 is the cost of the first unit produced

C_n is the cost of the nth unit produced

n is the number of units produced

r is the learning factor

This program solves for any of the above variables and also solves for the average cost over a range from i to j using the formula:

$$\overline{C}_n = \frac{C_1}{j-i} \cdot \frac{j^{B+1} - i^{B+1}}{B+1}$$

where $B = \log r / \log 2$

The theory applies to a single product, or closely related series of similar products developing through the evolutionary process. The average cost is approximate because of the finite, discrete nature of the function. Small values of i may produce incorrect results.

Example:

A computer manufacturer begins a pilot run on a component. Cost accounting informs him that the first unit off of the line cost $975 and the 100th unit a week later costs $643. What cost can the manufacturer expect for the 10,000th unit of the line? What is the average cost of the 10,000 units?

Keystrokes:	Display:	
[USER]		(Set USER Mode)
[XEQ] [ALPHA] SIZE [ALPHA] 008		
[XEQ] [ALPHA] LEARN [ALPHA]	0.00	
975 [A]	COST1=975.00	
100 [D]	N=100.00	
643 [C]	COSTN=643.00	
[B]	R=0.94	(Learning factor)
10000 [D]	N=10,000.00	
[C]	COSTN=424.05	(10,000 unit cost)
1 [ENTER↑]	1.00	
10000 [E]	AVG$=466.13	(Average for 10,000 units)

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program and set USER mode		[USER]	
2.	Initialize		[XEQ] LEARN	0.00
3.	Input three of the following:			
	cost of the first unit	C_1	[A]	COST1=
	learning factor	r	[B]	R=
	cost of the nth unit	C_n	[C]	COST N=
	number of units	n	[D]	N=
4.	Compute the remaining variable:			
	cost of the first unit		[A]	COST 1=
	learning factor		[B]	R=
	cost of the nth unit		[C]	COST N=
	number of units		[D]	N=
5.	Compute the average cost from the ith			
	to the jth unit	i	[ENTER↑]	
		j	[E]	AVG$=

Program Listings

```
 01◆LBL "LEA
RN"
 02 CF 22
 03 FIX 2
 04 SF 21
 05 CLX
 06 RTN
 07◆LBL A
 08 "COST1"
 09 STO 01
 10 FS?C 22
 11 GTO 02
 12 RCL 02
 13 XEQ 01
 14 RCL 04
 15 X<>Y
 16 Y↑X
 17 RCL 03
 18 X<>Y
 19 /
 20 STO 01
 21 GTO 02
 22 RTN
 23◆LBL B
 24 "R"
 25 STO 02
 26 FS?C 22
 27 GTO 02
 28 RCL 04
 29 XEQ 00
 30 STO 02
 31 GTO 02
 32◆LBL 00
 33 XEQ 01
 34 RCL 03
 35 RCL 01
 36 /
 37 LOG
 38 X<>Y
 39 /
 40 10↑X
 41 RTN
 42◆LBL C
 43 "COSTN"
 44 STO 03
 45 FS?C 22
 46 GTO 02
 47 RCL 02
 48 XEQ 01
 49 RCL 04
 50 X<>Y
```

Initialize

Calculate Cost 1 (C_1)

Calculate r

Calculate Cost n (C_n)

```
 51 Y↑X
 52 RCL 01
 53 *
 54 STO 03
 55 GTO 02
 56◆LBL D
 57 "N"
 58 STO 04
 59 FS?C 22
 60 GTO 02
 61 RCL 02
 62 XEQ 00
 63 2
 64 1/X
 65 +
 66 INT
 67 STO 04
 68 GTO 02
 69◆LBL 01
 70 LOG
 71 2
 72 LOG
 73 /
 74 RTN
 75◆LBL E
 76 "AVG$"
 77 STO 05
 78 RCL 02
 79 XEQ 01
 80 1
 81 +
 82 STO 07
 83 Y↑X
 84 X<>Y
 85 STO 06
 86 RCL 07
 87 Y↑X
 88 -
 89 RCL 07
 90 /
 91 RCL 01
 92 *
 93 RCL 05
 94 RCL 06
 95 -
 96 /
 97◆LBL 02
 98 "⊢="
 99 ARCL X
100 AVIEW
101 .END.
```

Calculate n

B=log r/log2

Calculate $\overline{C}_n$

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	C_l		
	r^l		
	Cn		
	n		
05	i ; Cn	55	
	j		
	B+1		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 031 USER MODE

ENG ____ FIX 2 SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
22		Digit entered	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

PRICE ELASTICITY OF DEMAND

Using historical (or estimated)prices and resulting unit sales, this program calculates the elasticity of demand (elasticity of quantity sold with respect to a change in price).

Equations:

$$E_d = \frac{\Delta Q}{\frac{1}{2}(Q_i + Q_{i+1})} \div \frac{\Delta P}{\frac{1}{2}(P_i + P_{i+1})}$$

where:

E_d = demand elasticity
Q_{i+1} = quantity sold after price change
Q_i = quantity sold before price change
P_{i+1} = new price
P_i = old price
i = 1,2,3,..., n
$\Delta Q = Q_{i+1} - Q_i$
$\Delta P = P_{i+1} - P_i$

$$\frac{\Delta P}{\frac{1}{2}[P_i + P_{i+1}]} \neq 0$$

Example:

The sales volume of a product varied with the different price changes per unit as follows:

N	Quantity sold (Q)	Price/Unit (P)
1	0	6
2	10	4
3	20	2
4	30	0*

Compute the price elasticity of demand.

*hypothetical price for simplicity

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 005	
[XEQ] [ALPHA] DEMAND [ALPHA]	PRICE=?
6 [R/S]	QUANTITY=?
0 [R/S]	PRICE=?
4 [R/S]	QUANTITY=?
10 [R/S]	Ed=5.0000
[R/S]	PRICE=?
2 [R/S] 20 [R/S]	Ed=1.0000
[R/S]	PRICE=?
0 [R/S] 30 [R/S]	Ed=0.2000

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program			
2.	Initialize		[XEQ] DEMAND	PRICE=?
3.	Key in first price of series	Price 1	[R/S]	QUANTITY=?
4.	Key in first quantity of series	Quantity1	[R/S]	PRICE=?
5.	Key in subsequent price	Price n	[R/S]	QUANTITY=?
6.	Key in subsequent quantity and calculate			
	demand elasticity	Quantityn	[R/S]	Ed=
7.	Press [R/S] to return to step 5 to enter			
	next price		[R/S]	PRICE=?
8.	For a new case, go to step 2.			

Program Listings

```
 01◆LBL "DEM
AND"
 02 SF 21
 03 XEQ 03
 04 STO 02
 05 XEQ 04
 06 STO 04
 07◆LBL 00
 08 XEQ 03
 09 X<> 02
 10 STO 01
 11 XEQ 04
 12 X<> 04
 13 STO 03
 14 4
 15 STO 00
 16◆LBL 01
 17 XEQ 02
 18 DSE 00
 19 GTO 01
 20 /
 21 CHS
 22 "Ed="
 23 ARCL X
 24 AVIEW
 25 GTO 00
 26◆LBL 02
 27 RCL IND
00
 28 ENTER↑
 29 DSE 00
 30 RCL IND
00
 31 -
 32 X<>Y
 33 LASTX
 34 +
 35 /
 36 RTN
 37◆LBL 03
 38 "PRICE=?
"
 39 PROMPT
 40 RTN
 41◆LBL 04
 42 "QUANTIT
Y=?"
 43 PROMPT
 44 .END.
```

Initialize

Shift data

Calculate E_d

Prompting subroutines

51			
60			
70			
80			
90			
00			

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	USED	50	
	P_i		
	P_{i+1}		
	Q_1		
	Q_{i+1}		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 005 TOT. REG. 18 USER MODE

ENG ____ FIX ____ SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21		Printer enable	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

NOTES

NOTES

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41C programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41C, HP has an extensive library of "Application Pacs". These programs transform your HP-41C into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, minature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41C.

You can choose from:

Aviation
Clinical Lab
Circuit Analysis
Financial Decisions
Mathematics
Structural Analysis
Surveying
Securities
Statistics
Stress Analysis
Games
Home Management
Machine Design
Navigation
Real Estate
Thermal and Transport Science

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41C Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

* Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales
Home Construction Estimating
Lending, Saving and Leasing
Real Estate
Small Business
Geometry
High-Level Math
Test Statistics
Antennas
Chemical Engineering
Control Systems
Electrical Engineering
Fluid Dynamics and Hydraulics
Civil Engineering
Heating, Ventilating & Air Conditioning
Mechanical Engineering
Solar Engineering
Calendars
Cardiac/Pulmonary
Chemistry
Games
Optometry I (General)
Optometry II (Contact Lens)
Physics
Surveying
Time Module Solutions I

* Some books require additional memory modules to accomodate all programs.

BUSINESS STATISTICS/MARKETING SALES

FORECASTING USING EXPONENTIAL SMOOTHING
MONTHLY SEASONAL VARIATION FACTORS BASED ON CENTERED MOVING AVERAGES
MULTIPLE LINEAR REGRESSION
NORMAL, INVERSE NORMAL, T AND F DISTRIBUTIONS
BASIC STATISTICS FOR TWO VARIABLES
MOVING AVERAGE
GOMPERTZ CURVE TREND ANALYSIS
BREAK-EVEN ANALYSIS
EXPERIENCE (LEARNING) CURVE FOR MANUFACTURING COST
PRICE ELASTICITY OF DEMAND

00041-90094 Rev. D 4/81 Printed in U.S.A.

HEWLETT-PACKARD

HP-41

USERS' LIBRARY SOLUTIONS

Business Stat/Marketing/Sales

Bar Codes

BUSINESS STAT/MARKETING/SALES

FORECASTING USING EXPONENTIAL
SMOOTHING
PROGRAM REGISTERS NEEDED: 39

ROW 1 (1 - 3)

ROW 2 (3 - 10)

ROW 3 (10 - 18)

ROW 4 (19 - 29)

ROW 5 (30 - 39)

ROW 6 (40 - 52)

ROW 7 (53 - 59)

ROW 8 (59 - 68)

ROW 9 (68 - 76)

ROW 10 (77 - 86)

ROW 11 (86 - 95)

ROW 12 (96 - 104)

ROW 13 (105 - 111)

ROW 14 (111 - 119)

ROW 15 (120 - 126)

ROW 16 (126 - 131)

ROW 17 (132 - 141)

ROW 18 (142 - 147)

ROW 19 (148 - 153)

ROW 20 (153 - 161)

ROW 21 (162 - 167)

MONTHLY SEASONAL VARIATION
FACTORS – CENTERED MOVING AVGS
PROGRAM REGISTERS NEEDED: 19

ROW 1 (1 – 6)

ROW 2 (7 – 10)

ROW 3 (11 – 17)

ROW 4 (18 – 28)

ROW 5 (29 – 41)

ROW 6 (42 – 54)

ROW 7 (55 – 67)

ROW 8 (67 – 73)

ROW 9 (74 – 78)

ROW 10 (78 – 81)

ROW 11 (82 – 82)

MULTIPLE LINEAR REGRESSION

PROGRAM REGISTERS NEEDED: 36

ROW 1 (1 – 4)

ROW 2 (5 – 15)

ROW 3 (16 – 24)

ROW 4 (25 – 33)

ROW 5 (34 – 43)

ROW 6 (43 – 50)

ROW 7 (51 – 61)

ROW 8 (62 – 71)

ROW 9 (72 – 84)

ROW 10 (85 – 97)

ROW 11 (98 – 110)

ROW 12 (111 – 123)

ROW 13 (124 – 136)

ROW 14 (136 – 142)

ROW 15 (143 – 153)

ROW 16 (154 – 166)

ROW 17 (167 – 175)

ROW 18 (176 – 183)

ROW 19 (183 - 190)

ROW 20 (190 - 190)

NORMAL AND INVERSE NORMAL
DISTRIBUTIONS
PROGRAM REGISTERS NEEDED: 44

ROW 1 (1 – 3)

ROW 2 (3 – 6)

ROW 3 (7 – 9)

ROW 4 (10 – 11)

ROW 5 (12 – 14)

ROW 6 (14 – 16)

ROW 7 (16 – 18)

ROW 8 (18 – 22)

ROW 9 (22 – 24)

ROW 10 (24 – 28)

ROW 11 (28 – 35)

ROW 12 (36 – 47)

ROW 13 (47 – 53)

ROW 14 (54 – 64)

ROW 15 (65 – 77)

ROW 16 (78 – 87)

ROW 17 (88 – 97)

ROW 18 (98 – 105)

ROW 19 (106 – 115)

ROW 20 (116 – 128)

ROW 21 (129 – 141)

ROW 22 (141 – 151)

ROW 23 (152 – 159)

ROW 24 (159 – 162)

T DISTRIBUTIONS

PROGRAM REGISTERS NEEDED: 35

ROW 1 (1 – 6)

ROW 2 (7 – 14)

ROW 3 (15 – 25)

ROW 4 (26 – 38)

ROW 5 (39 – 49)

ROW 6 (50 – 61)

ROW 7 (62 – 70)

ROW 8 (71 – 82)

ROW 9 (83 – 94)

ROW 10 (95 – 106)

ROW 11 (107 – 116)

ROW 12 (117 – 128)

ROW 13 (128 – 137)

ROW 14 (138 – 149)

ROW 15 (150 – 158)

ROW 16 (159 – 169)

ROW 17 (170 – 180)

ROW 18 (181 – 190)

ROW 19 (190 – 193)

F DISTRIBUTIONS

PROGRAM REGISTERS NEEDED: 20

ROW 1 (1 - 3)

ROW 2 (4 - 11)

ROW 3 (12 - 19)

ROW 4 (20 - 31)

ROW 5 (32 - 44)

ROW 6 (45 - 55)

ROW 7 (56 - 64)

ROW 8 (65 - 77)

ROW 9 (77 - 87)

ROW 10 (87 - 97)

ROW 11 (98 - 101)

BASIC STATISTICS FOR TWO
VARIABLES
PROGRAM REGISTERS NEEDED: 36

ROW 1 (1 - 4)

ROW 2 (5 - 14)

ROW 3 (15 - 24)

ROW 4 (25 - 34)

ROW 5 (34 - 41)

ROW 6 (41 - 51)

ROW 7 (51 - 55)

ROW 8 (56 - 60)

ROW 9 (61 - 69)

ROW 10 (69 - 75)

ROW 11 (75 - 81)

ROW 12 (82 - 88)

ROW 13 (89 - 96)

ROW 14 (96 - 104)

ROW 15 (104 - 111)

ROW 16 (112 - 120)

ROW 17 (120 - 129)

ROW 18 (129 - 135)

ROW 19 (136 - 140)

MOVING AVERAGE

PROGRAM REGISTERS NEEDED: 44

ROW 1 (1 - 3)

ROW 2 (3 - 7)

ROW 3 (8 - 14)

ROW 4 (15 - 18)

ROW 5 (18 - 26)

ROW 6 (26 - 35)

ROW 7 (35 - 40)

ROW 8 (41 - 50)

ROW 9 (51 - 55)

ROW 10 (55 - 59)

ROW 11 (59 - 62)

ROW 12 (62 - 72)

ROW 13 (73 - 81)

RCW 14 (82 - 92)

ROW 15 (92 - 94)

ROW 16 (95 - 101)

ROW 17 (102 - 109)

ROW 18 (110 - 116)

MOVING AVERAGE

ROW 19 (116 - 119)

ROW 20 (119 - 129)

ROW 21 (130 - 138)

ROW 22 (139 - 145)

ROW 23 (146 - 147)

ROW 24 (147 - 150)

GOMPERTZ CURVE TREND ANALYSIS

PROGRAM REGISTERS NEEDED: 20

ROW 1 (1 – 6)

ROW 2 (7 – 14)

ROW 3 (15 – 22)

ROW 4 (23 – 33)

ROW 5 (34 – 46)

ROW 6 (47 – 59)

ROW 7 (60 – 72)

ROW 8 (73 – 79)

ROW 9 (80 – 87)

ROW 10 (88 – 97)

ROW 11 (97 – 98)

BREAK–EVEN ANALYSIS

PROGRAM REGISTERS NEEDED: 32

ROW 1 (1 – 4)

ROW 2 (5 – 7)

ROW 3 (7 – 9)

ROW 4 (9 – 11)

ROW 5 (11 – 13)

ROW 6 (13 – 16)

ROW 7 (17 – 25)

ROW 8 (25 – 35)

ROW 9 (36 – 41)

ROW 10 (42 – 48)

ROW 11 (49 – 58)

ROW 12 (58 – 67)

ROW 13 (68 – 73)

ROW 14 (73 – 79)

ROW 15 (80 – 85)

ROW 16 (85 – 93)

ROW 17 (94 – 100)

EXPERIENCE(LEARNING) CURVE
FOR MANUFACTURING COST
PROGRAM REGISTERS NEEDED: 24

ROW 1 (1 - 3)

ROW 2 (4 - 9)

ROW 3 (10 - 18)

ROW 4 (19 - 27)

ROW 5 (27 - 34)

ROW 6 (35 - 43)

ROW 7 (43 - 50)

ROW 8 (51 - 59)

ROW 9 (60 - 68)

ROW 10 (69 - 76)

ROW 11 (77 - 87)

ROW 12 (88 - 98)

ROW 13 (99 - 101)

PRICE ELASTICITY OF DEMAND

PROGRAM REGISTERS NEEDED: 14

ROW 1 (1 - 3)

ROW 2 (3 - 9)

ROW 3 (10 - 17)

ROW 4 (18 - 24)

ROW 5 (25 - 33)

ROW 6 (34 - 39)

ROW 7 (40 - 42)

ROW 8 (43 - 44)

NOTES

NOTES

NOTES

00041-90094 Rev. D 4/81 Printed in U.S.A.